


«Solitaire» for the Commodore 64

(Complete type-in listing in BASIC V2)



**New
Game**

KLONDIKE SOLITAIRE



WELCOME TO SOLITAIRE
FOR COMMODORE MOUSE 1351 IN PORT 2

F1 PLAY RANDOM
F3 PLAY NUMBER
D DRAW 1 CARD
C SHOW COUNTER
S SOUND ON



**An exclusive 2021 Give-Away by
Roman Werner (🐦 @RomWer)**

Thanks! for following me on Twitter

!- Note: Curly brackets {} represent special screen code characters (BASTEXT PETSCII Tokens)
!- For BASTEXT reference see: https://www.c64-wiki.com/wiki/PETSCII_Codes_in_Listings
!- For checksum tool see : <https://www.c64-wiki.com/wiki/F64Summer>

```
CSUM
11E2 10 rem *****
868F 20 rem *** klondike - solitaire ***
D03C 30 rem *** for commodore mouse 1351 ***
85BB 40 rem *** (c) 2021 by @romwer ***
70C5 50 rem *** roman.werner@gmail.com ***
1EC1 60 rem *****
E08D 70 poke53280,5:poke53281,5:print"{clear}{black}"spc(14)"please wait":dw=1:sh=1:so=7
A50C 80 fori=0to23:poke54272+i,0:next:poke54296,so:gosub1900:gosub2080:gosub1830
E175 90 rem *** title screen ***
E39C 100 print"{clear}{black}{reverse off}{down}"spc(11)"klondike solitaire{down*6}"
0E97 110 ifhithenfori=1to4:print"{home}{down*3}"spc(5+i*5)pt$(0);:next
600D 120 print:printspc(10)"{down}{white}welcome to solitaire{down}"
6B59 130 printspc(3)"{light green}for commodore mouse 1351 in port 2{down*2}"
4454 140 printspc(13)"{yellow}f1 {light green}play random{down}":dw$=left$("s",dw-1)
DAEA 150 printspc(13)"{yellow}f3 {light green}play number{white}";:ifsd>0thenprintsd;
8D5C 160 print:printspc(14)"{down}{yellow}d {light green}draw{white}"dw"{light green}card"dw$
5334 170 printspc(14)"{down}{yellow}c {white}"mid$("hideshow",sh*4+1,4)" {light green}counter"
0E3F 180 printspc(14)"{down}{yellow}s {light green}sound {white}"mid$("offon",so/2+1,3)
C275 190 poke251,pc:poke252,dw:poke254,so:poke2,sh:poke45,0:poke46,64:clr
1FCD 200 pc=peek(251):dw=peek(252):so=peek(254):sh=peek(2)
62DB 210 gosub1640:fori=1to4:print"{home}{down*3}"spc(5+i*5)pt$(0);:next:poke198,0
E6CF 220 poke214,16:print:printspc(28):ifk$="{f3}"ory=8thenprintss$sr$r$"{up*2}";
9533 230 getk$:if(peek(j2)and1)=0thengosub1490:waitj2,1:goto230
D6F5 240 y=0:x=0:if(peek(j2)andg)=0thenx=peek(v):y=peek(v+1)-53:y=int(y/16):waitj2,g
5BFC 250 ifk$=""andy=0goto230
1C5D 260 ifx>0thenifx<122goto230
FC02 270 ifk$="{f1}"ory=7thensd=int(rnd(1)*ti)+1:goto340
DA43 280 ifk$="{f3}"ory=8thenpoke19,1:input"{white}";a$:poke19,0:sd=val(a$):on-(sd>0)goto340
4A9D 290 ifk$="d"ory=9thendw=4-dw:poke214,18:print:printspc(21)cd$(dw)
4654 300 ifk$="c"ory=10thensh=1-sh:poke214,20:print:printspc(16)sh$(sh)
D6F8 310 ifk$="s"ory=11thensso=7-so:pokes+24,so:poke214,22:print:printspc(22)v$(so)
5535 320 poke646,1:goto220
BBB8 330 rem *** new game / shuffling
4429 340 a=rnd(-sd):mv=0:poke19,0:print"{clear}{black}"spc(15)"shuffling":pokes+4,129
B5C1 350 fori=1to15:next:pokes+4,0:fori=0tohi:cd(i)=i:next:fori=hi+1step-1
0430 360 a=rnd(1)*(i+1):b=cd(i):cd(i)=cd(a):cd(a)=b:ifi<30thenpokes+4,129:pokes+4,0
E01E 370 next
603B 380 rem *** show side panel
E6F1 390 print"{white}{clear}"spc(35)o$(0)d1$"{red}!"d$"{red}#"d$"{black}%"d$"{black}&"d2$u$(0)r$;
B4A5 400 print"{white}{down}"nc$r$"{down*8}{light green} @{reverse on} {reverse off}"/;
8238 410 rem *** lay out cards ***
E9F6 420 z=n:fori=0to6:lc(i)=n:next:ox=n:os=n:lc(7)=12:ifshthenpoke1422,48
5A23 430 fory=0to6:forx=yto6:z=z+1:cd=cd(z):ifx>ythencd=cd+128
5445 440 c(x,y)=cd:gosub710:next:lc(y)=y:next:e=z:pokes+4,129:pokes+4,0
019E 450 poke214,9:print:printspc(35)o$(0)pt$u$(2):ti$="000000"
DC2F 460 rem *****
1D26 470 rem *** main loop ***
FDD3 480 rem *****
2004 490 pokefe,0:cs=n:y2=n:waitfe,17:x=peek(fb)+msb(peek(fd))-23
6A96 500 x2=int(x/f):ifx2<0orx2>7orx-x2*f>30thenpokes+1:x2=n:goto630
C44F 510 rb=peek(fe)and1:lc=lc(x2):of=of(x2):y=peek(fc)-53:y2=int(y/g)+of
8EBA 520 iflc=nthenifos=12oros=25oros=38oros=51thenpokes+1:y2=n:gosub1170
A54A 530 ify2>lc+2orlc=nthenpokes+1:os=n:y2=n:goto630
95DD 540 ify2>0thenify2>lctheny2=lc:ify>lc*g+fthenpokes+1:x2=n:y2=n:goto630
15A3 550 ifx2=7thengosub830:goto630
1EB6 560 ifx2=oxthenify2=oythenifos>nthenonrbgosub910:goto640
74ED 570 pokes+1:ify2=12thenifof(x2)=0thengosub1370:os=n:goto490
598E 580 ify2=8thenifof(x2)thengosub1370:os=n:goto490
5411 590 ify2>nthencs=c(x2,y2):ifcs>hiandy2<lcthencs=n
283E 600 ifcs>hithenify2=lcthencd=cs-128:c(x2,y2)=cd:gosub700:x2=n:cs=n
E9E0 610 ifos>nthenifcs>nthenifx2<>oxthenify2=lcthengosub1140
1CCB 620 ifcs>nthengosub1270
9580 630 ox=x2:oy=y2:os=cs:ifrbthenifos>nandox>nthengosub910
365B 640 ifshthenifmvthenmv$=str$(mv):print"{home}{down*9}{white}"spc(35)right$(" "+mv$,4)
5DAD 650 waitj2,g:waitj2,1:goto490
30F1 660 rem *****
816D 670 rem *** subroutines ***
295F 680 rem *****
2E39 690 rem *** draw card on screen
BF38 700 x=x2:y=y2
E7C3 710 tb=x*5:pokes+4,129:u=0
AA30 720 of=of(x):print"{home}"left$(ln$,y+y-of-of)spc(tb)
B2BD 730 ifcd>hithenprintpt$(sgn(y));:pokes+4,0:return
CC36 740 a=int(cd/nc):r=cd-a*nc-1:ifr<0thenr=12
```

```

C6C7 750 f$=str$(r+2):ifr>8thenf$=" "+mid$("jqka",r-8,1)
8B72 760 e$=d$(a):c$=c$(a):g$=mid$(f$+" ",2,2):f$=right$(" "+f$,2)
F438 770 o=0:ifyandx<7thento=1:ifc(x,y-1)>hi thento=2
FE7B 780 ify-of<10thenprinto$(o)v$e$f$o$w$q$g$e$w$u$(u);:pokes+4,0:return
BDDC 790 ify-of<11thenprinto$(o)v$e$f$o$w$q$g$e$w$;:pokes+4,0:return
E30A 800 ify-of<12thenprinto$(o)v$e$f$o$w$q$;:pokes+4,0:return
6F45 810 printo$(o);:pokes+4,0:return
F395 820 rem *** handle side panel click
F354 830 ifx2=nory2=nthenreturn
8D31 840 ify2>4andy2<8andy<121thenife<hiorz<hi thengosub1000:goto880
B95E 850 ify2>7andy2<11andy<168thengoto1100
5DF7 860 ify2>nandy2<4thenifos>nthengosub910
5BEA 870 ify2>10andy2<13thenify>187andx>283andx<303thenpoke$e,1:waitj2,16:goto100
62AC 880 poke$e,1:cs=n:x2=n
D0DC 890 return
9061 900 rem *** foundation pile update
7AE0 910 a=int(os/nc):r=os-a*nc-1:iffp(a)<>roryoy<lc(ox)thenreturn
6413 920 ifr<0thenr=12
305F 930 poke$e,1:fp(a)=fp(a)+1:f$=str$(r+2):ifr>8thenf$=" "+mid$("jqka",r-8,1)
AF7B 940 g$=mid$(f$+" ",2,2):print"{home}"left$(ln$,a*2+1)spc(35){white}.c$(a)g$
FB63 950 ifox=7thengosub1220:os=n:oy=n
ADF7 960 ifox<7thenlc(ox)=lc(ox)-1:gosub1410:os=n
10AA 970 mv=mv+1:ifz=hiand e=z thenwn=0:fori=0to6:wn=wn+lc(i):next:ifwn=-7thengoto1510
C710 980 pokes+4,129:pokes+4,0:return
88F5 990 rem *** stock pile
CB77 1000 ifrbthengoto1490
BEA9 1010 poke$e,1:forj=1todw
26AC 1020 print"{home}"left$(ln$,10)spc(35):os=n:ox=n:u=hi-e-1:ifu>2thenu=2
3A82 1030 ifz=hi thenz=e:printo$(0)pt$u$(u)b$b$b$b$b$b$;:mv=mv+1:return
A82F 1040 pokes+4,129:z=z+1:cd=cd(z):ifz=hi thenprintnc$;
444B 1050 ifz=hi-2orz=hi-1thenu=hi-z-1:printo$(0)pt$u$(u);
96C8 1060 u=z-e-1:ifu>2thenu=2
C2C5 1070 print"{home}"left$(ln$,g)spc(35):y=0:gosub740:u=0:pokes+4,0:ifz<hi thennext
880B 1080 return
ECB2 1090 rem *** waste pile
C583 1100 ifz=eorox=7andoy>8andoy<12orx2=oxandy2=oy thencs=os:return
167C 1110 poke$e,1:pokev+g,peek(v+g)or130:pokev+2,39:pokev+3,178:pokev+14,39
6D01 1120 pokev+15,184:poke$e,131:cs=cd(z):x2=7:y2=12:c(7,12)=cs:return
CE12 1130 rem *** check if selected card goes under clicked card
D162 1140 ifcs<26andos<26oros>25andcs>25thenreturn
936B 1150 cv=int(cs/nc):cv=cs-cv*nc:ov=int(os/nc):ov=os-ov*nc
D614 1160 ifcv<=ovorcv-ov>1thenreturn
8906 1170 fori=0tolc(ox)-oy:c(x2,lc(x2)+i+1)=c(ox,oy+i):next:lc(x2)=lc(x2)+i-1
F257 1180 ifox<7thenlc(ox)=lc(ox)-i:gosub1410
5B2D 1190 ifox=7thengosub1220
074C 1200 gosub1470:c(x2,y2+1)=os:lc(x2)=lc(x2)+1:cs=n:x2=n:return
1BEC 1210 rem *** take a card off the discard pile ***
BDDE 1220 x=ox:y=8:tb=35:e=e+1:u=z-e-1:ifu>2thenu=2
071F 1230 ifz=e thencd=cd(z):print"{home}"left$(ln$,g)spc(35)s$b$b$b$b$b$b$;
8341 1240 ifz>e thencd=cd(z-1):gosub720:fori=ztoe+1step-1:cd(i)=cd(i-1):next
791B 1250 return
D983 1260 rem *** show selector frame
F9FB 1270 xp=x2*f+15:yp=(y2-of)*g+50:pokev+2,xp:pokev+3,yp
5B43 1280 lc=lc(x2):ifof=0andlc>12thenlc=12
7A35 1290 dv=lc-y2:sp=0:pokev+14,xp:pokev+15,(lc-of)*g+56:ifdv<3goto1350
16CD 1300 ifdv>2thenpokev+4,xp:pokev+5,yp+36:sp=4
9D05 1310 ifdv>4thenpokev+6,xp:pokev+7,yp+68:sp=12
011D 1320 ifdv>6thenpokev+8,xp:pokev+9,yp+100:sp=28
7F62 1330 ifdv>8thenpokev+10,xp:pokev+11,yp+132:sp=60
F627 1340 ifdv>10thenpokev+12,xp:pokev+13,yp+164:sp=124
2207 1350 poke$e,131orsp:pokev+g,peek(v+g)and1:return
CFD4 1360 rem *** redraw whole column (offset change)
E767 1370 x=x2:y=y2:of(x)=9-of(x):tb=x*5:print"{home}"spc(tb)s$;
FB53 1380 fory=0to5:printb$b$b$b$b$;:next:for y=of(x)tolc(x):cd=c(x,y):gosub710:next
95A1 1390 oy=n:return
45DB 1400 rem *** remove selected cards from source column
FFED 1410 x=ox:y=oy:u=0:tb=x*5:ify=0thenprint"{home}"spc(tb)s$b$b$b$b$b$b$;:goto1440
ED5D 1420 y=y-1:ify<of(x)thengosub1370:return
13CC 1430 cd=c(x,y):gosub720
D18F 1440 a=24-(y-of(x))*2-6:ifa<0thenreturn
0CEB 1450 fori=0toa:printb$;:next:return
8CD5 1460 rem *** add selected cards to the target column
3CCA 1470 x=x2:for y=y2+1tolc(x2)+1:cd=c(x,y):gosub710:next:mv=mv+1:return
0F7F 1480 rem *** change card deck color
C91D 1490 pc=pc+1and3:pokev+36,p(pc):fori=0to6:pokev+40+i,p(pc):next:return
0F24 1500 rem *** show statistics
D1A9 1510 t$=ti$:print"{home}{reverse off}"spc(13){black}congratulations"
4EAF 1520 print"{light green}game time {white}"left$(t$,2)"h "mid$(t$,3,2)"m "right$(t$,2)"s"
192A 1530 print"{light green}seed number{white}"sd:print"{light green}card moves {white}"mv:mv=0

```

```

C6B7 1540 print"{light green}draw mode {white}"dw:waitj2,16:pokefe,0:c=12:y=0:x=0:of=0:u=0
CE0F 1550 printtab(34)"{up}"s$:fori=0to12:forj=0to3:cd=c+j*nc:tb=j*5+14:pokeys+4,129
9D60 1560 print"{home}"left$(ln$,rnd(0)*10+10)spc(rnd(0)*31)
72AD 1570 b=peek(fe):hi=51:ifbthenpokefe,0:hi=n:g$=""
2982 1580 gosub730:print"{home}"left$(ln$,j*2+1)spc(36)c$g$:next:c=c-1:next
9050 1590 print"{home}{down}"spc(36)"{light gray}{reverse on}";:fori=0to3
AE4E 1600 print"@ {down*2}{left*2}";:next:return
E778 1610 rem *****
8D66 1620 rem *** initialization ***
989D 1630 rem *****
8767 1640 n=-1:f=40:g=16:hi=51:nc=13:msb(1)=256:v=53248:s=54272:j2=56320:se=v+21
4AEB 1650 fb=251:fc=252:fd=253:p(0)=14:p(1)=10:p(2)=12:p(3)=4
DD16 1660 ln$="{down*24}":r$="{left*5}{down}":o$="{reverse off}":rd$="{reverse on}{red}"
4762 1670 bl$="{reverse on}{black}":c$(0)=rd$:c$(1)=rd$:c$(2)=bl$:c$(3)=bl$
EA03 1680 q$=r$+"{white}.{reverse on}{space*3}{reverse off}"/":d$(0)=rd$+"!"
521A 1690 d$(1)=rd$+"#":d$(2)=bl$+"%":d$(3)=bl$+"&":s$="{space*5}":v$=r$+"{white}."
E6CB 1700 w$="{white}"/":b$=r$+s$:v$(0)="off":v$(7)="on ":o$(0)="{white}()")*"
AD1A 1710 o$(1)="{white}:{cm f*3}{white}<":sh$(0)="hide":sh$(1)="show"
A3CE 1720 o$(2)="{white}:{reverse on}{white}{cm @*3}{reverse off}{white}<"
B9AB 1730 u$(0)=r$+"{white}+,,-":u$(1)=r$+"{white}[£££]":u$(2)=r$+"{white}=>>>?"
F5BF 1740 n$=r$+".{reverse on}{white}{cm m*3}{reverse off}"+w$:pt$=n$+n$+n$+n$
5097 1750 pt$(0)=o$(0)+pt$+u$(0):pt$(1)=o$(2)+pt$+u$(0):cd$(1)="{white}1 {light green}card "
51CB 1760 cd$(3)="{white}3 {light green}cards":d1$=r$+"{white}.{reverse on}{light gray}@ "
3F9B 1770 d2$="{white}{reverse off}"/":d3$=r$+"^";;{arrow left}":d$=d2$+d3$+d1$
9CFD 1780 nc$="{light green}() )"+r$+".{space*3}"/"+b$b$+r$+".{space*3}"/"+r$+"+, , -"
7683 1790 dimi,j,x,y,x2,y2,cs,ox,oy,os,lc,cd(hi),c(7,18)
3F5B 1800 pokev+34,15:pokev+35,1:pokeys,31:pokeys+1,31:fori=0to3:fp(i)=n:next
C3D2 1810 pokev+24,peek(v+24)and240or14:pokev+17,peek(v+17)or64:return
7177 1820 rem *** sprite selector frame setup
9BDC 1830 fori=3to57step3:poke896+i,6:poke898+i,3:poke960+i,6:poke962+i,3:next
9A1B 1840 poke896,3:poke897,255:poke898,254:poke899,7:poke900,255:poke901,255
4502 1850 poke1017,7:poke1018,255:poke1019,255:poke1020,3:poke1021,255:poke1022,254
4D7B 1860 pokev+29,254:pokev+23,254:pokev+27,124
0CC6 1870 fori=0to6:pokev+40+i,14:next:fori=0to5:poke2041+i,14:next:poke2047,15
E82D 1880 fori=0to6:pokev+2+i+i,255:pokev+3+i+i,52+i*30:next:return
E552 1890 rem *** copy char rom to ram at $3800 (14336)
C8C0 1900 poke56334,peek(56334)and254:poke1,peek(1)and251:poke781,2:poke782,255
7639 1910 poke90,0:poke91,209:poke88,0:poke89,57:sys41960:poke1,peek(1)or4
19B0 1920 poke56334,peek(56334)or1:fori=0to7:reada:poke14336+i,a:next:fori=0to167
193A 1930 reada:poke14552+i,a:next:fori=0to47:reada:poke14800+i,a:next:return
9F72 1940 rem *** custom character data
2639 1950 data 0,8,12,126,126,12,8,0,7,7,7,3,4,3,0,0,255,255,255,255,0,255,0,0,224
7F46 1960 data 224,224,192,32,192,0,0,7,7,7,0,7,7,7,224,224,224,0,224,224,224,224
BF86 1970 data 0,0,0,0,0,0,0,54,127,127,127,62,28,8,0,102,102,0,0,0,0,0,8,28,62
8456 1980 data 127,62,28,8,0,153,66,36,0,255,255,255,255,8,28,42,127,42,8,28,0,8,28
9709 1990 data 62,127,127,54,8,0,153,66,36,153,153,36,66,153,0,0,0,3,7,7,0,0,0,0
5A0F 2000 data 255,255,255,255,0,0,0,192,224,224,224,7,7,7,3,0,0,0,0,255,255,255
9392 2010 data 255,0,0,0,0,224,224,224,192,0,0,0,0,7,7,7,7,7,7,224,224,224,224
D6F7 2020 data 224,224,224,224,7,7,7,4,3,7,7,7,255,255,255,0,255,255,255,255,224,224
745D 2030 data 224,32,192,224,224,224,7,7,7,3,4,3,4,3,255,255,255,255,0,255,0,255
894C 2040 data 224,224,224,192,32,192,32,192
7CF8 2050 rem *****
ECB3 2060 rem *** mouse pointer setup ***
EF9D 2070 rem *****
CA35 2080 ad=49152:h=int(ad/256):rem install driver at $c000 (49152)
AE72 2090 fori=0to217:reada:pokead+i,a:next:fori=1to8:reada:pokead+a,h:next
9822 2100 fori=0to62:reada:poke832+i,a:next:v=53248:pokev+36,14:pokev+39,0
CF37 2110 pokev,90:pokev+1,190:pokev+16,0:poke2040,13:pokev+21,1:sysad:return
663E 2120 rem *** 1351 mouse driver (port 2)
A888 2130 rem 251=xpos / 252=ypos / 253=xpos high bit / 254=mouse click values
FF5A 2140 rem process mouse click(s) - then poke254,0 to wait for next click
0147 2150 data 120,169,13,141,20,3,169,192,141,21,3,88,96,165,254,208,21,173,0,220
4078 2160 data 73,255,41,17,240,12,133,254,173,0,208,133,251,173,1,208,133,252,173
0FFD 2170 data 25,212,160,0,32,184,192,140,42,192,164,253,208,11,172,0,208,192,19
4EC0 2180 data 208,4,224,0,208,42,24,109,0,208,141,0,208,168,138,105,0,41,1,77,16
2B81 2190 data 208,141,16,208,41,1,133,253,208,8,192,19,176,13,169,19,208,6,192,81
AA5A 2200 data 144,5,169,81,141,0,208,173,26,212,160,0,32,184,192,140,110,192,172,1
67CE 2210 data 208,192,50,208,4,224,0,240,35,192,247,208,4,224,0,208,27,56,73,255
5463 2220 data 109,1,208,201,50,176,8,201,30,144,8,169,50,208,6,201,247,144,2,169
7ED1 2230 data 247,141,1,208,169,192,72,169,176,72,8,72,72,72,76,49,234,169,191,141
F521 2240 data 0,220,76,129,234,140,194,192,141,216,192,162,0,56,233,0,41,127,201,64
F526 2250 data 176,5,74,208,11,138,96,9,192,201,255,240,248,56,106,202,160,0,96
A8DF 2260 rem *** driver hi byte patch table so it can be installed other than $c000
CCDA 2270 data 7,45,48,113,116,164,186,189
CA71 2280 rem *** mouse pointer sprite data
5A57 2290 data 6,0,0,9,0,0,9,0,0,9,0,0,9,192,0,9,56,0,9,38,0,9,37,0
90DC 2300 data 233,36,128,152,4,128,136,0,128,72,0,128,40,0,128,32,0,128
5C92 2310 data 16,0,128,16,1,0,8,1,0,8,1,0,4,2,0,3,252,0

```


!- For Joystick version please replace (or add) the following lines:

```
030E 30 rem *** for joystick ***
B604 130 printspc(9)"{light green}for joystick in port 2{down*2}"
4AC4 200 pc=peek(251):dw=peek(252):so=peek(254):sh=peek(2):poke254,0
092C 230 getk$:ifpeek(fe)and1thengosub1490:waitj2,g:pokefe,0:goto230
F7DD 630 ox=x2:oy=y2:os=cs:ifpeek(fe)and1thenifos>nandox>nthengosub910
E3F2 650 waitj2,g:goto490
276E 1000 rem ifrbthengoto1490
1C3A 2090 fori=0to254:reada:pokead+i,a:next:fori=1to15:reada:pokead+a,h:next
9822 2100 fori=0to62:reada:poke832+i,a:next:v=53248:pokev+36,14:pokev+39,0
CF37 2110 pokev,90:pokev+1,190:pokev+16,0:poke2040,13:pokev+21,1:sysad:return
CD32 2120 rem *** joystick driver
A888 2130 rem 251=xpos / 252=ypos / 253=xpos high bit / 254=mouse click values
FF5A 2140 rem process mouse click(s) - then poke254,0 to wait for next click
3181 2150 data 120,169,13,141,20,3,169,192,141,21,3,88,96,173,0,220,73,255,41,16,170
FCF1 2160 data 169,0,208,25,138,240,73,238,22,192,133,254,174,0,208,134,251,174,1
F473 2170 data 208,134,252,169,0,141,70,192,208,51,169,0,208,8,138,208,44,238,51,192
5E24 2180 data 208,39,169,0,208,21,238,70,192,169,0,201,20,176,15,138,240,23,238,63
2E8B 2190 data 192,9,1,133,254,208,14,138,208,11,169,0,141,22,192,141,51,192,141,63
1BB1 2200 data 192,162,1,173,0,220,41,15,73,15,208,10,160,1,140,102,192,160,10,140
C65B 2210 data 254,192,206,254,192,16,16,160,10,140,254,192,172,102,192,192,3,176,4
5235 2220 data 200,140,102,192,74,144,3,206,1,208,74,144,3,238,1,208,74,144,20,172,0
B9E6 2230 data 208,208,10,168,173,16,208,73,1,141,16,208,152,206,0,208,208,0,74,144
126F 2240 data 16,238,0,208,173,0,208,208,8,173,16,208,73,1,141,16,208,202,208,159
8B73 2250 data 169,50,205,1,208,144,3,141,1,208,169,247,205,1,208,176,3,141,1,208
ED80 2260 data 173,16,208,41,1,133,253,208,12,169,19,205,0,208,144,3,141,0,208,208
8081 2270 data 10,169,81,205,0,208,176,3,141,0,208,76,49,234,10
0154 2280 rem *** driver hi byte patch table so it can be installed other than $c000
6F68 2290 data 7,30,47,59,68,80,94,97,100,116,121,124,131,134,142
8B20 2300 rem *** mouse pointer sprite data
7314 2310 data 6,0,0,9,0,0,9,0,0,9,0,0,9,0,0,9,56,0,9,38,0,9,37,0
5E57 2320 data 233,36,128,152,4,128,136,0,128,72,0,128,40,0,128,32,0,128
F8CA 2330 data 16,0,128,16,1,0,8,1,0,8,1,0,4,2,0,3,252,0
```

!- add-on to auto-move cards to foundation pile at end of the game (trigger by right mouse button)
!- this is optional since it is also may slow the overall game loop down a tiny little bit

```
F765 511 ifrb=0ore<hithengoto520
CCDC 512 fori=1to6:on-(c(i,0)>hi)goto520:next
1A07 513 ford=0to6:ox=d:oy=lc(d):ifoy>nthenos=c(d,oy):gosub910
EC9B 514 next:on-(wn>-7)goto513:goto650
```

!- good to know...

- You can move cards directly to the foundation pile by clicking the right mouse button.
- When a tableau column gets too long on screen so that you can't see the bottom cards anymore then just click the very bottom card element in that column to have the complete bottom half of this column card displayed. Vice versa to return to the top half of the same column again, click the very top card element.
- You can change the card deck color by clicking the right mouse button on the stock pile.
- If you chose <F3> PLAY NUMBER by mistake then enter a "0" to cancel entry mode.
- When you want to restart a game then exit the game. Back at the menu choose <F3> PLAY NUMBER and without changing the number press <enter> to play the last game/same seed again.
- When you want to share a tricky, but solvable game with your friends then after you return to the menu, write down the seed number of your last game (the number that stands next to <F3> PLAY NUMBER) and tell your friends to enter and play this specific shuffle seed.